1983/43. A theoretical basis for gravity residual specification (Revision 1).

R.G. Richardson

Abstract

The aperture of the box-car filter required to produce a gravity residual corresponding to material within a specified depth range may be determined by comparing the theoretical gravity spectrum with the filter response.

INTRODUCTION

The interpretation of a gravity anomaly map normally requires the enhancement of particular features using a filtering process. The filtering may take the form of spectral truncation, running means, or visual smoothing of profiles. These techniques are frequently subjective, with the result that two interpreters rarely produce the same enhanced map.

It may be shown (Grant, 1974) that the phase angle of the spectrum of the gravity anomaly from one body depends only on the horizontal location of the body. The shape of the spectrum is determined by the body depth only. The spectrum of the anomaly from a body of finite dimensions declines more rapidly with increasing frequency than the spectrum from a point mass at the same depth of burial. Thus by calculating the spectrum of the anomaly from a point mass at a given depth, the maximum high frequency spectral content from any body at the same depth is defined.

APPLICATION

The normalised amplitude spectrum of the anomaly of a point mass at a depth z is

$$A = e^{-\pi zs}$$

where s is cycles per unit distance (Grant, 1974).

The high frequency cut-off is then defined as the point on the spectral curve at which the normalised amplitude is 0.707 (fig. 1). The high frequency cut-off at any desired body depth may be determined from a plot of this type.

The normalised amplitude response of a box-car low-pass filter is

$$A = \sin \frac{(D\pi s)}{D\pi s}$$

where D is the aperture
s is cycles per unit distance (Leaman and Richardson, 1981).

The high-frequency cut-off (fig. 2) is defined in the same way as the spectral cut-off. A plot for any desired aperture allows the cut-off to be measured.

It should be noted that by taking observations with finite spacing, the data is filtered with an aperture of the size of the nominal station spacing.

To select a filter aperture, plot the anomaly spectrum for a point mass at the maximum depth of interest and measure the cut-off frequency. Within the constraints imposed by the observation spacing, plot a series of filter responses and select the one with a cut-off closest to the desired cut-off. The plots may be made using FORTRAN program GRVRES (Appendix 1).

REFERENCES

GRANT, F.S. 1974. Geophysical data processing. *Unpublished course notes from Macquarie University.*

LEAMAN, D.E.; RICHARDSON, R.G. 1981. Gravity survey of the East Coast coalfields. *Bull.geol.Surv.Tasm.* 60.

RICHARDSON, R.G. 1983. Hard copy plotting on the Geological Survey mini-computer. *Unpubl.Rep.Dep.Mines Tasm.* 1983/38.

[8 September 1983]



Figure 1. *Anomaly spectrum.*



Figure 2. *Moving average.*

5 cm

APPENDIX 1

Program GRVRES

The program issues prompts on logical unit 3 and reads the plot parameters from logical unit 5.

Input is:

FI, AN or BO - format A2
- choose a plot of filter responses, a plot of anomaly spectra, or both

CYCMAX - free format
- the maximum number of cycles per kilometre to which the plots are to extend.

If filter responses are to be plotted:

AFILT - free format one per line
- up to 14 aperture sizes (in kilometres) terminated by a negative number

If spectra are to be plotted:

AFREQ - free format one per line
- up to 14 source depths (in kilometres) terminated by a negative number

Details of the plotting subroutines may be found in Richardson (1983).

APPENDIX 2

```
$TITL  GRVRES - PLOT MOVING AVERAGE FILTER RESPONSE OR ANOMALY SPECTRUM
      DIMENSION X(1000),Y(1000),AFILT(15),AFREQ(15)
      INTEGER*2 FI,AN,BO,IOPT,BUFF(33)
      LOGICAL FILT,DPTH,IER,IPN
      DATA FI/2HFI/,AN/2HAN/,BO/2HBO/
      OPEN(UNIT=3,FILE='CON:')
      OPEN(UNIT=5,FILE='CON:')
C OPEN THE CONSOLE FOR INPUT AND OUTPUT
      YMAX=0.
   10 WRITE(3,100)
  100 FORMAT(' (FI)LTER RESPONSE,(AN)OMALY SPECTRUM OR (BO)TH?')
      READ(5,101) IOPT
  101 FORMAT(A2)
      IF (IOPT .NE. FI .AND. IOPT .NE. AN .AND. IOPT .NE. BO) GOTO 10
C NEED VALID OPTION
      FILT=IOPT .EQ. FI .OR. IOPT .EQ. BO
      DPTH=IOPT .EQ. AN .OR. IOPT .EQ. BO
      CALL INITAL(9,200,36,0,0,0)
C INITIALLISE PLOTTER
      CALL FACTOR(0.3937)
C PLOT IN CENTIMETRES
      CALL PLOT(-5.,-5.,0)
C SET ORIGIN
      WRITE(3,102)
  102 FORMAT(' MAXIMUM NO. OF CYCLES/KM?')
      READ(5,*) CYCMAX
      NX=IFIX(CYCMAX*100.0+1.)
C NO OF STEPS IN CURVE
      IF (.NOT. FILT) GOTO 12
C DONT WANT FILTER
      IFILT=0
      WRITE(3,103)
  103 FORMAT(' ENTER APERTURE SIZES IN KM - TERMINATE BY NEGATIVE')
      DO 11 I=1,15
      READ(5,*) ANUM
      IF (ANUM .LT. 0.) GOTO 12
C END OF INPUT
      IFILT=IFILT+1
      AFILT(IFILT)=ANUM
   11 CONTINUE
C READ MOVING AVERAGE SIZES
C
C NOW FOR VARIATION OF ANOMALY SPECTRUM WITH DEPTH
   12 IF (.NOT. DPTH) GOTO 15
C DONT WANT FREQUENCY - DEPTH RELATION
      IFREQ=0
      WRITE(3,104)
  104 FORMAT(' ENTER DEPTH IN KM - TERMINATE BY NEGATIVE')
      DO 14 I=1,15
      READ(5,*) ANUM
      IF (ANUM .LT. 0.) GOTO 15
C END OF INPUT
      IFREQ=IFREQ+1
      AFREQ(IFREQ)=ANUM
   14 CONTINUE
C READ DEPTHS OF THE POINT SOURCE
   15 CONTINUE
      IF (.NOT. FILT) GOTO 19
```

```
C DONT WANT FILTER RESPONSE
      CALL PLOT(0.,YMAX,-3)
C SET ORIGIN
      YMAX=25.
C SAVE MAX VALUE IN Y DIRECTION FOR LATER
      ENCODE(BUFF,200)
  200 FORMAT('MOVING AVERAGE')
      CALL SYMBOL(2.,-4.,.28,BUFF,0.,14)
C LABEL FILTER GRAPH
C NOW FORM AXES
      CALL PAXIS(CYCMAX,-0.25,25.)
      DO 17 J=1,IFILT
      X(1)=0.
      Y(1)=1.
C SET THE FIRST VALUES
      DO 16 I=2,NX
      XTMP=FLOAT(I-1)*0.01
      X(I)=XTMP
      IF (AFILT(J) .GT. 0.) GOTO 18
      Y(I)=1.0
C IF NO FILTER HAVE ALL PASS
      GOTO 16
   18 CONTINUE
      Y(I)=SIN(AFILT(J)*3.141592*XTMP)/(AFILT(J)*3.141592*XTMP)
   16 CONTINUE
      CALL LDRAW(X,Y,NX,CYCMAX*10.,25.0,YMAX,-0.25,AFILT(J),J)
   17 CONTINUE
C NOW FOR THE -3 DB POINT WHICH IS THE CUT-OFF OF THE FILTER
      YVAL=0.707*20.0+5.0
      CALL CUTOFF(YVAL,NX)
   19 CONTINUE
C NOW FOR THE SPECTRUM OF A POINT SOURCE WITH INCREASING
C BURIAL
      IF (.NOT. DPTH) GOTO 22
C ONLY WANTED FILTER
      CALL PLOT(0.,YMAX+5.0,-3)
C SET ORIGIN AGAIN
      YMAX=20.
C UPPER LIMIT OF Y
      ENCODE(BUFF,201)
  201 FORMAT('ANOMALY SPECTRUM')
      CALL SYMBOL(2.,-2.,.28,BUFF,0.,16)
C LABEL PLOT
      CALL PAXIS(CYCMAX,0.,20.)
      DO 21 J=1,IFREQ
      X(1)=0.
      Y(1)=1.
C SET THE FIRST VALUES
      DO 20 I=2,NX
      XTMP=FLOAT(I-1)*0.01
      X(I)=XTMP
      Y(I)=EXP(-3.141592*AFREQ(J)*XTMP)
   20 CONTINUE
      CALL LDRAW(X,Y,NX,CYCMAX*10.,20.,YMAX,0.,AFREQ(J),J)
   21 CONTINUE
      YVAL=0.707*20.0
      CALL CUTOFF(YVAL,NX)
   22 CONTINUE
      CALL RSTR(2)
      STOP
```

```
      END
      SUBROUTINE LINE(X,Y,N,KODE,ISPACE)
      INTEGER*4 KODE,ISPACE,N,I,J
      DIMENSION X(N),Y(N)
      CALL PENUP
      I=KODE
      IF (KODE .LT. 0) I=-I
      DO 10 J=1,N
      CALL PLOT(X(J),Y(J),1)
      IF (MOD(J-1,ISPACE) .NE. 0) GOTO 10
      CALL FACTOR(1.)
      IF (KODE .NE. 0) CALL MARKER(I)
      CALL FACTOR(0.3937)
      IF (KODE .GE. 0) CALL PENDN
   10 CONTINUE
      CALL PENUP
      RETURN
      END
      SUBROUTINE CUTOFF(VAL,NX)
      LOGICAL IPN
      IPN=.TRUE.
      CALL PLOT(0.,VAL,3)
      CALL PENDN
      DO 10 I=6,NX,5
      XTMP=FLOAT(I-1)*0.01
      CALL PLOT(XTMP*10.,VAL,1)
      IPN=.NOT. IPN
      IF (.NOT. IPN) CALL PENUP
      IF (IPN) CALL PENDN
   10 CONTINUE
      RETURN
      END
      SUBROUTINE PAXIS(XMAX,YMIN,YLNTH)
      DIMENSION X(2),Y(2)
      INTEGER*2 BUFF(33)
      X(1)=0.
      Y(1)=1.
      X(2)=XMAX
      Y(2)=YMIN
      CALL SCALE(X,2,XMAX*10.,FMIN,DELTA)
      ENCODE(BUFF,100)
  100 FORMAT('CYCLES/KM')
      CALL AXIS(0.,0.,BUFF,-9,XMAX*10.,0.,FMIN,DELTA,1)
C DRAW AND SCALE X AXIS
      CALL SCALE(Y,2,YLNTH,FMIN,DELTA)
      ENCODE(BUFF,101)
  101 FORMAT('AMPLITUDE')
      CALL AXIS(0.,0.,BUFF,9,YLNTH,90.,FMIN,DELTA,2)
      RETURN
      END
      SUBROUTINE LDRAW(X,Y,NX,XMAX,YLNTH,YMAX,YMIN,VAL,J)
      DIMENSION X(2),Y(2),BUFF(33)
      X(NX+1)=XMAX*0.1
      Y(NX+1)=YMIN
C FORCE THE RANGE FOR SCALING
      CALL SCALE(X,NX+1,XMAX,FMIN,DELTA)
      CALL SCALE(Y,NX+1,YLNTH,FMIN,DELTA)
      CALL LINE(X,Y,NX,J,10)
      ENCODE(BUFF,200) VAL
  200 FORMAT(F5.2,' KM&')
```

```
CALL PLOT(XMAX+1.,YMAX-FLOAT(J+2),3)
CALL FACTOR(1.)
CALL MARKER(J)
CALL FACTOR(0.3937)
CALL PWRITE(XMAX+2.0,YMAX-FLOAT(J)-2.14,0.28,0.,BUFF)
RETURN
END
```